

Universidade Federal de Alagoas
Instituto de Computação
Bacharelado em Ciência da Computação

Especificação da Gramática - Isengard++

Márcio Henrique Vieira de Oliveira
Michael Miller Rodrigues Cardoso

Maceió - AL, 2021

Sumário

Sumário	2
1. Tipo de Analisador Sintático	3
2. Gramática livre de contexto	4
3. Gramática LL(1)	6

1. Tipo de Analisador Sintático

O tipo de analisador sintático escolhido para o presente trabalho foi o analisador descendente LL(1) preditivo recursivo.

2. Gramática livre de contexto

S = DcMain | DcFun S | Dcld S | ϵ

DcMain = 'Funct' 'Int' 'Main' '(' ' ' BlockDc

DcFun = 'Funct' FunType 'id' '(' Param ')' BlockDc

FunType = 'Void' | VarType

VarType = 'Int' | 'Float' | 'Char' | 'Str' | 'Bool'

Param = ParamDc | ϵ

ParamDc = VarType 'id' Vet ',' ParamDc | VarType 'id' Vet

BlockDc = 'Begin' Instructions 'End'

Vet = '[' ']' | ϵ

Instructions = Dcld Instructions | Command Instructions | CommandIO Instructions | FunCall Instructions | AtrDir Instructions | Return Instructions | ϵ

Dcld = VarType DcldAtr ';'

DcldAtr = Id | Id ',' DcldAtr | Atr | Atr ',' DcldAtr

Id = 'id' '[' Ea ']' | 'id'

Atr = 'id' '[' Ea ']' '=' '[' AtrVet ']' | 'id' '=' Ec

AtrVet = Ec ',' AtrVet | Ec

AtrDir = 'id' '[' Ea ']' '=' Ec ';' | 'id' '=' Ec ';'

Command = IfElse | While | For

CommandIO = Input | Output

FunCall = 'id' '(' ParamFun ')' ';'

ParamFun = ParamFun ',' Ec | Ec | ϵ

Return = 'Return' Ec ';'

IfElse = 'If' '(' Eb ')' BlockDc | 'If' '(' Eb ')' BlockDc 'Else' BlockDc

While = 'While' '(' Eb ')' BlockDc

For = 'For' '(' DcInt ',' IntValue ',' IntValue ')' BlockDc | 'For' '(' DcInt ',' IntValue ')' BlockDc

DcInt = 'Int' 'id' | 'id'

IntValue = 'id' | 'CT_INT'

Input = 'Input' '(' InputParam ')' ';'

InputParam = 'id' | 'id' ',' InputParam

Output = 'Output' '(' OutputParam ')' ';' | 'Outputln' '(' OutputParam ')' ';'

OutputParam = Ec | Ec ',' OutputParam

Ec = Ec 'OP_CONCAT' Eb | Eb

Eb = Eb 'PR_OR' Tb | Eb 'PR_AND' Tb | Tb

Tb = Tb 'PR_NOT' Ra | Ra

Ra = Ra Rel Rb | Rb

Rb = Rb Ops Ea | Ea

Ea = Ea 'OP_AD' Ta | Ea 'OP_SUB' Ta | Ta

Ta = Ta 'OP_MULT' Fa | Ta 'OP_DIV' Fa | Ta 'OP_RES' Fa | Fa

Fa = '(' Ec ')' | Id | FunCall | 'CT_INT' | 'CT_FLOAT' | 'PR_TRUE' | 'PR_FALSE' | 'CT_CHAR' | 'CT_STR' | 'OP_NOTUNI' 'id' | 'OP_SIZE' 'id'

Rel = 'OP_RELEQUAL' | 'OP_RELDIF'

Ops = 'OP_GREATER' | 'OP_LESS' | 'OP_GREATERT' | 'OP_LESST'

3. Gramática LL(1)

S = 'Funct' DcFun S | DcId S | ϵ

DcFun = FunTypeMinusInt 'id' '(' Param ')' BlockDc | 'Int' DcIntFunMain

FunTypeMinusInt = 'Void' | 'Float' | 'Char' | 'Str' | 'Bool'

DcIntFunMain = 'Main' '(' ')' BlockDc | 'id' '(' Param ')' BlockDc

VarType = 'Int' | 'Float' | 'Char' | 'Str' | 'Bool'

Param = ParamDc | ϵ

ParamDc = VarType 'id' Vet ParamDcFat

ParamDcFat = ',' ParamDc | ϵ

BlockDc = 'Begin' Instructions 'End'

Vet = '[' ']' | ϵ

Instructions = DcId Instructions | Command Instructions | CommandIO Instructions | 'id' AtrDirFunCall Instructions | Return Instructions | ϵ

DcId = VarType DcIdAtr ';'

DcIdAtr = 'id' Id Atr DcIdAtrFat

DcIdAtrFat = ',' DcIdAtr | ϵ

Id = '[' Ea ']' | ϵ

Atr = '=' AtrFat | ϵ

AtrFat = '[' AtrVet ']' | Ec

AtrVet = Ec AtrVetFat

AtrVetFat = ',' AtrVet | ϵ

Command = If | While | For

CommandIO = Input | Output

AtrDirFunCall = AtrDir | FunCall

AtrDir = '[' Ea ']' '=' Ec ';' | '=' Ec ';' ;

FunCall = '(' FunCallParam

FunCallParam = ')' ';' | ParamFun ')' ';' ;

ParamFun = Ec ParamFunLL

ParamFunLL = ';' ParamFun | ε

IdFunCall = Id | FunCall

Return = 'Return' ReturnParam

ReturnParam = ';' | Ec ';' ;

If = 'If' '(' Eb ')' BlockDc Else

Else = 'Else' BlockDc | ε

While = 'While' '(' Eb ')' BlockDc

For = 'For' '(' Dclnt ';' IntValue ForFat

ForFat = ';' IntValue ')' BlockDc | ')' BlockDc

Dclnt = 'Int' 'id' '=' IntValue | 'id' AtrInt

IntValue = 'id' | 'CT_INT'

AtrInt = '=' IntValue | ε

Input = 'Input' '(' InputParam ')' ';' ;

InputParam = 'id' Id InputParamFat

InputParamFat = ';' InputParam | ε

Output = 'Output' '(' OutputParam ')' ';' | 'Outputln' '(' OutputParam ')' ';' ;

OutputParam = Ec OutputParamFat

OutputParamFat = ',' OutputParam | ϵ

Ec = Eb EcLL

EcLL = 'OP_CONCAT' Eb EcLL | ϵ

Eb = Tb EbLL

EbLL = 'PR_OR' Tb EbLL | 'PR_AND' Tb EbLL | ϵ

Tb = Ra TbLL

TbLL = 'PR_NOT' Ra TbLL | ϵ

Ra = Rb RaLL

RaLL = Rel Rb RaLL | ϵ

Rb = Ea RbLL

RbLL = Ops Ea RbLL | ϵ

Ea = Ta EaLL

EaLL = 'OP_AD' Ta EaLL | 'OP_SUB' Ta EaLL | ϵ

Ta = Fa TaLL

TaLL = 'OP_MULT' Fa TaLL | 'OP_DIV' Fa TaLL | 'OP_RES' Fa TaLL | ϵ

Fa = '(' Ec ')' | 'id' IdFunCall | 'CT_INT' | 'CT_FLOAT' | 'PR_TRUE' | 'PR_FALSE' | 'CT_CHAR' | 'CT_STR' | 'OP_NOTUNI' 'id' | 'OP_SIZE' 'id'

Rel = 'OP_RELEQUAL' | 'OP_RELDIF'

Ops = 'OP_GREATER' | 'OP_LESS' | 'OP_GREATERT' | 'OP_LESST'